Лабораторная работа №2

Численное решение нелинейных систем

Задание. Найти объемные расходы Q_i жидкости на участках сложного трубопровода. Диаметры d и абсолютные шероховатости Δ всех труб одинаковые. Транспортируемая жидкость имеет кинематическую вязкость $\nu = 10^{-6}$ м^2/с. Варианты заданий приведены в таблице.

№ вар.	Схема трубопровода	L_1, L_2, … м	d, мм Δ, мм	$\zeta_{м1}, \zeta_{м2}$, …	Q_A, Q_B, … м^3/с
1		110 70 110 70 110	280 0.80	1.5 1.1 0.3 0.3	0.1 0.5 0.4
2		50 70 50 40 40	350 0.7	0.5 1 1.5 0 1	0.3 0.2 0.3 0.1
3		60 80 60 80	250 0.80	0.5 1.2 1.8 0 1	0.2 0.1 0.3
4		70 50 70 50	250 0.9	0.75 1 0 0.6 0.8	0.2 0.4 0.2

5		80 50 60 80 100 100	450 1	1 1 0 1.2 0.7	0.3 0.3 0.6
6		90 90 140 140	400 1.2	0.5 0 0.5 0.9 1.2	0.25 0.4 0.35
7		100 130 80 100 130 80 80	300 1.1	0.5 1 0 1 0.5	0.2 0.3 0.1
8		110 50 70 110 70 90 70	300 1.3	0 0.5 0.5 1 0.5 0 0	0.4 0.2 0.3
9		120 60 60 120 90 160 90	250 0.5	1 0 1.5 0 1 0 1	0.1 0.2 0.2
10		130 150 170 130 150 170 100 100	200 1	0 0 0.5 1 1.5 0 0 1	0.2 0.3 0.1 0.25

11		60 60 70 120 70 -	300 0.8	1.2 0.5 0.8 0.5 1.1 0.3	0.1 0.2 0.2
12		100 100 150 200 150 -	200 1.2	1 0.8 0.6 1.5 0.8 0.6	0.3 0.2 0.2
13		60 80 60 50 50	250 1.2	0.5 1 1.5 0 1 0.4	0.1 0.5 0.2 0.3
14		100 80 100 40 40	220 0.80	2.4 1.2 0.1 0.2 0.9 0.3	0.3 0.4 0.3
15		60 80 60 50 50	180 1.6	0.6 1.1 1.3 0.2 1.3 0.7	0.2 0.2 0.35 0.45
16		90 90 140 140 - 100 120	350 1.6	0.5 0 0.5 0.9 1.2	0.35 0.3 0.25 0.7

17		40 40 60 40 40 60 45 55	200 0.8	0.1 0.5 0.5 1.0 0.6 0 0.25 0.45	0.4 0.2 0.7
18		60 80 60 50 50 80 -	160 1.2	0.3 1.5 1.2 0. 1.0 0.3 0.5	0.3 0.25 0.15 0.50
19		60 60 70 120 70 - -	240 1.6	1.0 0.0 0.8 0.6 1.1 0.3 0.7	0.3 0.1 0.1
20		130 150 170 130 150 170 100	200 1	0.3 0.2 0.0 1.3 1.0 0 0.6	0.3 0.2 0.1 0.2
21		110 50 70 110 80 -	180 1.1	0.2 0.0 0.6 1.2 0.3 0 0.1	0.3 0.4 0.1 0.15

22		50 70 50 40 40 70 70	150 1.0	0.4 1.4 1.3 0.1 1.2 0.5 0.7	0.45 0.15 0.35 0.40
23		100 80 100 80	300 0.60	2 1 0 0.5	0.3 0.4 0.3

У к а з а н и я . Расходы Q_i (м^3/с) на участках трубопровода должны удовлетворять системе уравнений:

1. Баланс расходов в узлах:



$$Q_i + Q_j = Q_k$$

Формулируем $N-1$ таких уравнений, где N – общее количество узлов.

2. Баланс напоров, т.е. равенство нулю алгебраической суммы потерь напора для каждого кольца при подсчете по направлению движения часовой стрелки или против нее. Потери напора считаются положительными при совпадении направлений обхода контура и течения жидкости.

Например, для трубопровода с $N=5$ узлами и 2 кольцами



система уравнений имеет вид

$$Q_1 - Q_2 = Q_A, \quad Q_2 + Q_6 - Q_3 = Q_B, \quad Q_4 + Q_3 = Q_C, \quad Q_5 - Q_4 - Q_6 = 0,$$

$$\Delta H_1 + \Delta H_2 - \Delta H_6 - \Delta H_5 = 0, \quad \Delta H_3 - \Delta H_4 + \Delta H_6 = 0. \tag{1}$$

Здесь $\Delta H_i = K_i Q_i |Q_i|$ – потери напора (м), $K_i = \beta_i \zeta_i$, $\beta_i = 8/(\pi^2 g d_i^4)$, $\zeta_i = \zeta_{\text{м}i} + \lambda(Q_i, d_i, \Delta_i)\dfrac{l_i}{d_i}$ – коэффициент сопротивления i-го участка; $\lambda(Q, d, \Delta) = 0.11\big(\Delta/d + 68/(cQ)\big)^{0.25}$ – коэффициент гидравлического трения, $c = 4/(\pi \nu d)$, ($cQ \equiv \mathrm{Re}$ – число Рейнольдса); $\zeta_{\text{м}i}$ – коэффициент местных сопротивлений. С учетом этого (1) принимает вид

$$\begin{bmatrix} Q_1 - Q_2 \\ Q_2 - Q_3 + Q_6 \\ Q_3 + Q_4 \\ -Q_4 + Q_5 - Q_6 \\ K_1 Q_1|Q_1| + K_2 Q_2|Q_2| - K_5 Q_5|Q_5| - K_6 Q_6|Q_6| \\ K_3 Q_3|Q_3| - K_4 Q_4|Q_4| + K_6 Q_6|Q_6| \end{bmatrix} = \begin{bmatrix} Q_A \\ Q_B \\ Q_C \\ 0 \\ 0 \\ 0 \end{bmatrix},$$

или в матрично-векторной форме

$$A(\vec{Q}) \cdot \vec{Q} = \vec{b}, \tag{2}$$

где

$$A(\vec{Q}) = \begin{bmatrix} 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 1 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 & 1 & -1 \\ K_1|Q_1| & K_2|Q_2| & 0 & 0 & -K_5|Q_5| & -K_6|Q_6| \\ 0 & 0 & K_3|Q_3| & -K_4|Q_4| & 0 & K_6|Q_6| \end{bmatrix},$$

$$\vec{Q} = \begin{bmatrix} Q_1 \\ Q_2 \\ Q_3 \\ Q_4 \\ Q_5 \\ Q_6 \\ \\ \end{bmatrix}, \quad \vec{b} = \begin{bmatrix} Q_A \\ Q_B \\ Q_C \\ 0 \\ 0 \\ 0 \end{bmatrix}.$$

Для решения системы нелинейных уравнений (2) применяем метод последовательных приближений с инерцией (3.16) по следующему алгоритму:

1) Задаем весовой коэффициент α ($0 < \alpha < 1$, рекомендуемое значение 0.7…0.8), точность ε и максимальное число итераций N_{it}.

2) Задаем расходы на участках на нулевой и первой итерациях:

$$\vec{Q}_i^{(0)} = Q_0 / N, \quad \vec{Q}_i^{(1)} = \vec{Q}_i^{(0)}, \; i = 1,...,N,$$

где $Q_0 = Q_A + Q_B + Q_C$ – расход в магистрали.

3) Для $k = 1,2,...,N_{it}$

решаем СЛАУ

$$A((1-\alpha)\vec{Q}^{(k-1)} + \alpha\vec{Q}^{(k)}) \cdot \vec{Q}^{(k+1)} = \vec{b}. \tag{3}$$

если $\max\limits_{1 \le i \le N} \left| \vec{Q}_i^{(k+1)} - \vec{Q}_i^{(k)} \right| < \varepsilon$, то приближенное решение найдено: $\vec{Q}_i \approx \vec{Q}_i^{(k)}$, STOP.

Ниже приведена реализация алгоритма на языке Фортран. Для решения СЛАУ (3) используется метод Гаусса (процедура «Gauss»). Задание элементов матрицы $A(\vec{Q})$ системы осуществляется в процедуре «Matr», определение коэффициентов сопротивления и гидравлического трения – в процедурах «dzefun» и «alamfun» соответственно. В качестве примера рассчитывается изображенный выше трубопровод при $Q_A = 0.3$ м^3/с, $Q_B =$ 0.3 м^3/с, $Q_C = 0.6$ м^3/с, $d_1 = 120$ мм, $d_2 = 100$ мм, $d_3 = 150$ мм, $d_4 = 100$ мм, $d_5 = 150$ мм, $d_6 = 200$ мм, $\Delta = 1$ мм, $l_1 = 100$ м, $l_2 = 200$ м, $l_1 = 100$ м, $l_1 = 120$ м, $l_1 = 120$ м, $l_1 = 100$ м, $\zeta_{м1} = 10$, $\zeta_{м2} = 3$, $\zeta_{м3} = 1$, $\zeta_{м4} = 4$, $\zeta_{м5} = 1$, $\zeta_{м6} = 5$.

```
! Расчет сложного трубопровода -
! решение нелинейной системы A(q)*q=b
! методом послед. приближений с инерцией
!  A((1-alfa)*q_(k-2)+alfa*q_(k-1))*q_k=b
program hydro1
parameter(N=6) ! число участков !
dimension A(N,N),q(N),b(N),d(N),qold(N),qold2(N)
common /par1/bet(6),dzem(6),del(6),al(6),c(6)
pi=4*atan(1.)
anu=1.e-6                                     ! вязкость жидкости
d=(/120.,100.,150.,100.,150.,200./)/1e3 ! диаметры участков !
bet=8/(9.81*pi**2*d**4); bet=bet/bet(1)
c=4./(pi*anu*d)                               ! c*q=Re
dzem=(/10.,3.,1.,4.,1.,5./)                   ! к-ты местных сопротивлений
del=1.0e-3/d                                  ! относит. шероховатости !
al=(/100.,200.,100.,120.,120.,100./)/d        ! относит. длины !
qa=0.3; qb=0.3 ; qc=0.6                       ! расходы потребителей !
q0=qa+qb+qc                                   ! расход в магистрали !
b=(/qa,qb,qc,0.,0.,0./)                       ! правые части !
q=q0/N;  qold=q                               ! начальные приближения
print*,' q=',q
call Matr(N,q,A); Rmax=maxval(abs(matmul(A,q)-b))
print*,' Max. residual=',Rmax
eps=1.0e-6; alfa=0.8
it=0;  itmax=100
do
 qold2=qold
 qold=q
 call Matr(N,(1-alfa)*qold2+alfa*qold,A)
 call Gauss(N,A,b,q)
 it=it+1
 dq=maxval(abs(q-qold))
 print*,' Iter. #',it,' dqmax=',dq
 if(dq<eps) exit
 if(it==itmax) exit
end do
print*,' q=',q
call Matr(N,q,A); Rmax=maxval(abs(matmul(A,q)-b))
print*,' Max. residual=',Rmax
end

subroutine Matr(N,q,A) ! матрица A(q) !
dimension A(N,N),q(N),dze(N),aK(N)
common /par1/bet(6),dzem(6),del(6),al(6),c(6)
do i=1,N;dze(i)=dzefun(dzem(i),del(i),al(i),q(i),c(i));end do
aK=bet*dze
a(1,:)=(/1.,-1.,0.,0.,0.,0./)
a(2,:)=(/0.,1.,-1.,0.,0.,1./)
a(3,:)=(/0.,0.,1.,1.,0.,0./)
a(4,:)=(/0.,0.,0.,-1.,1.,-1./)
a(5,:)=(/ aK(1)*abs(q(1)) , aK(2)*abs(q(2)), 0., 0., &
         -aK(5)*abs(q(5)) ,-aK(6)*abs(q(6)) /)
a(6,:)=(/ 0., 0., aK(3)*abs(q(3)), -aK(4)*abs(q(4)), &
          0., aK(6)*abs(q(6)) /)
end

function dzefun(dzem,delt,al,q,c) ! к-т сопротивления
dzefun=dzem+alamfun(delt,q,c)*al
end

function alamfun(delt,q,c) ! к-т гидравлического трения
alamfun=0.11*(delt+68./(c*q))**0.25
end
```

Результаты:

```
q=  0.2000000      0.2000000      0.2000000      0.2000000      0.2000000
0.2000000
Max. residual=   7.001080
Iter. #           1 dqmax=  0.6977809
Iter. #           2 dqmax=  0.3686751
Iter. #           3 dqmax=  0.2535474
Iter. #           4 dqmax=  0.1163523
Iter. #           5 dqmax=  4.4093907E-02
Iter. #           6 dqmax=  1.1804163E-02
Iter. #           7 dqmax=  6.2674284E-04
Iter. #           8 dqmax=  1.8576980E-03
Iter. #           9 dqmax=  1.6092658E-03
Iter. #          10 dqmax=  9.1582537E-04
Iter. #          11 dqmax=  4.1043758E-04
Iter. #          12 dqmax=  1.4507771E-04
Iter. #          13 dqmax=  3.3974648E-05
Iter. #          14 dqmax=  1.7881393E-06
Iter. #          15 dqmax=  8.2254410E-06
Iter. #          16 dqmax=  6.2584877E-06
Iter. #          17 dqmax=  3.2782555E-06
Iter. #          18 dqmax=  1.3709068E-06
Iter. #          19 dqmax=  4.7683716E-07
q=  0.4040863      0.1040862      0.4319116      0.1680884      0.7959138
0.6278254
Max. residual=  3.3033509E-06
```

Как видим, для достижения заданной точности $\varepsilon = 10^{-6}$ потребовалось 19 итераций, расходы (м^3/с) жидкости на участках равны $Q_1 = 0.404$, $Q_2 = 0.104$, $Q_3 = 0.432$, $Q_4 = 0.168$, $Q_5 = 0.796$ и $Q_6 = 0.628$.